Lesson No. 14

Stack

Stack is a data structure that behaves in a first in last out manner. It can contain many elements and there is only one way in and out of the container. When an element is inserted it sits on top of all other elements and when an element is removed the one sitting at top of all others is removed first. To visualize the structure consider a test tube and put some balls in it. The second ball will come above the first and the third will come above the second. When a ball is taken out only the one at the top can be removed. The operation of placing an element on top of the stack is called pushing the element and the operation of removing an element from the top of the stack is called popping the element. The last thing pushed is popped out first; the last in first out behavior. 

We can peek at any ball inside the test tube but we cannot remove it without removing every ball on top of it. Similarly we can read any element from the stack but cannot remove it without removing everything above it. The stack operations of pushing and popping only work at the top of the stack. This top of stack is contained in the SP register. The physical address of the stack is obtained by the SS:SP combination. The stack segment registers tells where the stack is located and the stack pointer marks the top of stack inside this segment. 

Whenever an element is pushed on the stack SP is decremented by two as the 8088 stack works on word sized elements. Single bytes cannot be pushed or popped from the stack. Also it is a decrementing stack. Another possibility is an incrementing stack. A decrementing stack moves from higher addresses to lower addresses as elements are added in it while an incrementing stack moves from lower addresses to higher addresses as elements are added. There is no special reason or argument in favor of one or another, and more or less depends on the choice of the designers. Another processor 8051 by the same manufacturer has an incrementing stack while 8088 has a decrementing one. 

Memory is like a shelf numbered as zero at the top and the maximum at the bottom. If a decrementing stack starts at shelf 5, the first item is placed in shelf 5, the next item is placed in shelf 4, the next in shelf 3 and so on. The operations of placing items on the stack and removing them from there are called push and pop. The push operation copies its operand on the stack, while the pop operation makes a copy from the top of the stack into its operand. When an item is pushed on a decrementing stack, the top of the stack is first decremented and the element is then copied into this space. With a pop the element at the top of the stack is copied into the pop operand and the top of stack is incremented afterwards.

The basic use of the stack is to save things and recover from there when needed. For example we discussed the shortcoming in our last example that it destroyed the caller’s registers, and the callers are not supposed to remember which registers are destroyed by the thousand routines they use. Using the stack the subroutine can save the caller’s value of the registers on the stack, and recover them from there before returning. Meanwhile the subroutine can freely use the registers. From the caller’s point of view if the registers contain the same value before and after the call, it doesn’t matter if the subroutine used them meanwhile.

Similarly during the CALL operation, the current value of the instruction pointer is automatically saved on the stack, and the destination of CALL is loaded in the instruction pointer. Execution therefore resumes from the destination of CALL. When the RET instruction is executed, it recovers the value of the instruction pointer from the stack. The next instruction executed is therefore the one following the CALL. Observe how playing with the instruction pointer affects the program flow. 

There is a form of the RET instruction called “RET n” where n is a numeric argument. After performing the operation of RET, it further increments the stack pointer by this number, i.e. SP is first incremented by two and then by n. Its function will become clear when parameter passing is discussed.

Now we describe the operation of the stack in CALL and RET with an example. The top of stack stored in the stack pointer is initialized at 2000. The space above SP is considered empty and free. When the stack pointer is decremented by two, we took a word from the empty space and can use it for our purpose. The unit of stack operations is a word. Some instructions push multiple words; however byte pushes cannot be made. Now the value 017B is stored in the word reserved on the stack. The RET will copy this value in the instruction pointer and increment the stack pointer by two making it 2000 again, thereby reverting the operation of CALL. 

This is how CALL and RET behave for near calls. There is also a far version of these functions when the target routine is in another segment. This version of CALL takes a segment offset pair just like the far jump instruction. The CALL will push both the segment and the offset on the stack in this case, followed by loading CS and IP with the values given in the instruction. The corresponding instruction RETF will pop the offset in the instruction pointer followed by popping the segment in the code segment register.

Apart from CALL and RET, the operations that use the stack are PUSH and POP. Two other operations that will be discussed later are INT and IRET. Regarding the stack, the operation of PUSH is similar to CALL however with a register other than the instruction pointer. For example “push ax” will push the current value of the AX register on the stack. The operation of PUSH is shown below.

SP ( SP – 2
[SP] ( AX

The operation of POP is the reverse of this. A copy of the element at the top of the stack is made in the operand, and the top of the stack is incremented afterwards. The operation of “pop ax” is shown below.

AX ( [SP]
SP ( SP + 2

Making corresponding PUSH and POP operations is the responsibility of the programmer. If “push ax” is followed by “pop dx” effectively copying the value of the AX register in the DX register, the processor won’t complain. Whether this sequence is logically correct or not should be ensured by the programmer. For example when PUSH and POP are used to save and restore registers from the stack, order must be correct so that the saved value of AX is reloaded in the AX register and not any other register. For this the order of POP operations need to be the reverse of the order of PUSH operations. 

Now we consider another example that is similar to the previous examples, however the code to swap the two elements has been extracted into another subroutine, so that the formation of stack can be observed during nested subroutine calls.

	; bubble sort subroutine using swap subroutine

[org 0x0100]                     

              jmp start

data:         dw   60, 55, 45, 50, 40, 35, 25, 30, 10, 0

data2:        dw   328, 329, 898, 8923, 8293, 2345, 10, 877, 355, 98

              dw   888, 533, 2000, 1020, 30, 200, 761, 167, 90, 5

swapflag:     db   0

swap:         mov  ax, [bx+si]        ; load first number in ax

              xchg ax, [bx+si+2]      ; exchange with second number

              mov  [bx+si], ax        ; store second number in first

              ret                     ; go back to where we came from

bubblesort:   dec  cx                 ; last element not compared

              shl  cx, 1              ; turn into byte count

mainloop:     mov  si, 0              ; initialize array index to zero

              mov  byte [swapflag], 0 ; reset swap flag to no swaps

innerloop:    mov  ax, [bx+si]        ; load number in ax

              cmp  ax, [bx+si+2]      ; compare with next number

              jbe  noswap             ; no swap if already in order

              call swap               ; swaps two elements

              mov  byte [swapflag], 1 ; flag that a swap has been done

noswap:       add  si, 2              ; advance si to next index

              cmp  si, cx             ; are we at last index

              jne  innerloop          ; if not compare next two

              cmp  byte [swapflag], 1 ; check if a swap has been done   

              je   mainloop           ; if yes make another pass 

              ret                     ; go back to where we came from

start:        mov  bx, data           ; send start of array in bx

              mov  cx, 10             ; send count of elements in cx

              call bubblesort         ; call our subroutine

              mov bx, data2           ; send start of array in bx

              mov cx, 20              ; send count of elements in cx

              call bubblesort         ; call our subroutine again

              mov  ax, 0x4c00         ; terminate program

              int  0x21

	11

13
	A new instruction XCHG has been introduced. The instruction swaps its source and its destination operands however at most one of the operands could be in memory, so the other has to be loaded in a register. The instruction has reduced the code size by one instruction.

The RET at the end of swap makes it a subroutine.


Inside the debugger observe the use of stack by CALL and RET instructions, especially the nested CALL.

1.1. Saving and Restoring Registers

The subroutines we wrote till now have been destroying certain registers and our calling code has been carefully written to not use those registers. However this cannot be remembered for a good number of subroutines. Therefore our subroutines need to implement some mechanism of retaining the callers’ value of any registers used. 

The trick is to use the PUSH and POP operations and save the callers’ value on the stack and recover it from there on return. Our swap subroutine destroyed the AX register while the bubblesort subroutine destroyed AX, CX, and SI. BX was not modified in the subroutine. It had the same value at entry and at exit; it was only used by the subroutine. Our next example improves on the previous version by saving and restoring any registers that it will modify using the PUSH and POP operations. 

	; bubble sort and swap subroutines saving and restoring registers

[org 0x0100]                     

              jmp start

data:         dw   60, 55, 45, 50, 40, 35, 25, 30, 10, 0

data2:        dw   328, 329, 898, 8923, 8293, 2345, 10, 877, 355, 98

              dw   888, 533, 2000, 1020, 30, 200, 761, 167, 90, 5

swapflag:     db   0

swap:         push ax                 ; save old value of ax

              mov  ax, [bx+si]        ; load first number in ax

              xchg ax, [bx+si+2]      ; exchange with second number

              mov  [bx+si], ax        ; store second number in first

              pop  ax                 ; restore old value of ax

              ret                     ; go back to where we came from

bubblesort:   push ax                 ; save old value of ax

              push cx                 ; save old value of cx

              push si                 ; save old value of si

              dec  cx                 ; last element not compared

              shl  cx, 1              ; turn into byte count

mainloop:     mov  si, 0              ; initialize array index to zero

              mov  byte [swapflag], 0 ; reset swap flag to no swaps

innerloop:    mov  ax, [bx+si]        ; load number in ax

              cmp  ax, [bx+si+2]      ; compare with next number

              jbe  noswap             ; no swap if already in order

              call swap               ; swaps two elements

              mov  byte [swapflag], 1 ; flag that a swap has been done

noswap:       add  si, 2              ; advance si to next index

              cmp  si, cx             ; are we at last index

              jne  innerloop          ; if not compare next two

              cmp  byte [swapflag], 1 ; check if a swap has been done   

              je   mainloop           ; if yes make another pass 

              pop  si                 ; restore old value of si

              pop  cx                 ; restore old value of cx

              pop  ax                 ; restore old value of ax

              ret                     ; go back to where we came from

start:        mov  bx, data           ; send start of array in bx

              mov  cx, 10             ; send count of elements in cx

              call bubblesort         ; call our subroutine

              mov  bx, data2          ; send start of array in bx

              mov  cx, 20             ; send count of elements in cx

              call bubblesort         ; call our subroutine again

              mov  ax, 0x4c00         ; terminate program

              int  0x21

	19-21
	When multiple registers are pushed, order is very important. If AX, CX, and SI are pushed in this order, they must be popped in the reverse order of SI, CX, and AX. This is again because the stack behaves in a Last In First Out manner.


Inside the debugger we can observe that the registers before and after the CALL operation are exactly identical. Effectively the caller can assume the registers are untouched. By tracing into the subroutines we can observe how their value is saved on the stack by the PUSH instructions and recovered from their before exit. Saving and restoring registers this way in subroutines is a standard way and must be followed.

PUSH

PUSH decrements SP (the stack pointer) by two and then transfers a word from the source operand to the top of stack now pointed to by SP. PUSH often is used to place parameters on the stack before calling a procedure; more generally, it is the basic means of storing temporary data on the stack.

POP

POP transfers the word at the current top of stack (pointed to by SP) to the destination operand and then increments SP by two to point to the new top of stack. POP can be used to move temporary variables from the stack to registers or memory.

Observe that the operand of PUSH is called a source operand since the data is moving to the stack from the opernad, while the operand of POP is called destination since data is moving from the stack to the operand.

CALL

CALL activates an out-of-line procedure, saving information on the stack to permit a RET (return) instruction in the procedure to transfer control back to the instruction following the CALL. For an intrasegment direct CALL, SP is decremented by two and IP is pushed onto the stack. The target procedure’s relative displacement from the CALL instruction is then added to the instruction pointer. For an intersegment direct CALL, SP is decremented by two, and CS is pushed onto the stack. CS is replaced by the segment word contained in the instruction. SP again is decremented by two. IP is pushed onto the stack and replaced by the offset word in the instruction.

The out-of-line procedure is the temporary division, the concept of roundabout that we discussed. Near calls are also called intrasegment calls, while far calls are called intersegment calls. There are also versions that are called indirect calls; however they will be discuss later when they are used.

RET

RET (Return) transfers control from a procedure back to the instruction following the CALL that activated the procedure. RET pops the word at the top of the stack (pointed to by register SP) into the instruction pointer and increments SP by two. If RETF (intersegment RET) is used the word at the top of the stack is popped into the IP register and SP is incremented by two. The word at the new top of stack is popped into the CS register, and SP is again incremented by two. If an optional pop value has been specified, RET adds that value to SP. This feature may be used to discard parameters pushed onto the stack before the execution of the CALL instruction. 

